

# NAG Fortran Library Routine Document

## F07GRF (CPPTRF/ZPPTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07GRF (CPPTRF/ZPPTRF) computes the Cholesky factorization of a complex Hermitian positive-definite matrix, using packed storage.

### 2 Specification

```
SUBROUTINE F07GRF(UPLO, N, AP, INFO)
ENTRY      cpptrf (UPLO, N, AP, INFO)
INTEGER   N, INFO
complex AP(*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine forms the Cholesky factorization of a complex Hermitian positive-definite matrix  $A$  either as  $A = U^H U$  if UPLO = 'U', or  $A = LL^H$  if UPLO = 'L', where  $U$  is an upper triangular matrix and  $L$  is lower triangular, using packed storage.

### 4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates whether the upper or lower triangular part of  $A$  is stored and how  $A$  is factorized, as follows:

if UPLO = 'U', then the upper triangular part of  $A$  is stored and  $A$  is factorized as  $U^H U$ , where  $U$  is upper triangular;

if UPLO = 'L', then the lower triangular part of  $A$  is stored and  $A$  is factorized as  $LL^H$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

3: AP(\*) – *complex* array *Input/Output*

**Note:** the dimension of the array AP must be at least  $\max(1, N * (N + 1) / 2)$ .

*On entry:* the  $n$  by  $n$  Hermitian positive-definite matrix  $A$ , packed by columns. More precisely, if UPLO = 'U', the upper triangle of  $A$  must be stored with element  $a_{i,j}$  in AP( $i + j(j - 1)/2$ ) for  $i < j$ ; if UPLO = 'L', the lower triangle of  $A$  must be stored with element  $a_{i,j}$  in AP( $i + (2n - j)(j - 1)/2$ ) for  $i \geq j$ .

*On exit:* the upper or lower triangle of  $A$  is overwritten by the Cholesky factor  $U$  or  $L$  as specified by UPLO, using the same packed storage format as described above.

4: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the leading minor of order  $i$  is not positive-definite and the factorization could not be completed. Hence  $A$  itself is not positive-definite. This may indicate an error in forming the matrix  $A$ . To factorize a Hermitian matrix which is not positive-definite, call F07PRF (CHPTRF/ZHPTRF) instead.

## 7 Accuracy

If UPLO = 'U', the computed factor  $U$  is the exact factor of a perturbed matrix  $A + E$ , where

$$|E| \leq c(n)\epsilon|U^H||U|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

If UPLO = 'L', a similar statement holds for the computed factor  $L$ . It follows that  $|e_{ij}| \leq c(n)\epsilon\sqrt{a_{ii}a_{jj}}$ .

## 8 Further Comments

The total number of real floating-point operations is approximately  $\frac{4}{3}n^3$ .

A call to this routine may be followed by calls to the routines:

F07GSF (CPPTRS/ZPPTRS) to solve  $AX = B$ ;

F07GUF (CPPCON/ZPPCON) to estimate the condition number of  $A$ ;

F07GWF (CPPTRI/ZPPTRI) to compute the inverse of  $A$ .

The real analogue of this routine is F07GDF (SPPTRF/DPPTRF).

## 9 Example

To compute the Cholesky factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}.$$

using packed storage.

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07GRF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=8)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER       UPLO
*      .. Local Arrays ..
complex        AP(NMAX*(NMAX+1)/2)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        cpptrf, X04DDF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07GRF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
*
*      Factorize A
*
      CALL cpptrf(UPLO,N,AP,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Print factor
*
        IFAIL = 0
        CALL X04DDF(UPLO,'Nonunit',N,AP,'Bracketed','F7.4','Factor',
+                'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
      ELSE
        WRITE (NOUT,*) 'A is not positive-definite'
      END IF
      END IF
      STOP
*
      END

```

## 9.2 Program Data

F07GRF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
(3.23, 0.00)
(1.51, 1.92) ( 3.58, 0.00)
(1.90,-0.84) (-0.23,-1.11) ( 4.09, 0.00)
(0.42,-2.50) (-1.18,-1.37) ( 2.33, 0.14) ( 4.29, 0.00) :End of matrix A

```

### 9.3 Program Results

F07GRF Example Program Results

```
Factor
      1          2          3          4
1 ( 1.7972, 0.0000)
2 ( 0.8402, 1.0683) ( 1.3164, 0.0000)
3 ( 1.0572,-0.4674) (-0.4702, 0.3131) ( 1.5604, 0.0000)
4 ( 0.2337,-1.3910) ( 0.0834, 0.0368) ( 0.9360, 0.9900) ( 0.6603, 0.0000)
```

---